****************************************************************************
* File-Name: 		study2.do
* Date:		 12/19/2022
* Author: 		Batista Pereira et al.
* Purpose: 		Analysis of panel data for Study 2
* Data used: 		study2.dta
* Data Output:	
   * Table 2: itt and cace for intervention in study 2
   * Tables A16-A30 in Appendices 14-21
 ** Note: whenever possible we included codes to replicate all tables in .tex 
 *         format. The formatting of the replicated tables might differ  
 *         from the tables included in the manuscript. 
****************************************************************************

* For generating tables in .tex: 
ssc install estout, replace
ssc install tabout, replace
ssc install eclplot, replace
ssc install ivreg2, replace
ssc install ranktest, replace 
ssc install esizereg, replace 
ssc install leebounds, replace 


*************** RECODING VARIABLES

* MAIN TREATMENT

gen treat = treatment

* COMPLIANCE: OPENED EMAIL

gen comply = read_stimuli

gen comply2 = comply

recode comply2 (.=0)

* DROPPED AND SELF-SELECTED

gen select = w2_consent

recode select (.=0)

gen dropped = 1 - select

* BELIEF IN FAKE NEWS (1 is true, 2 is false)

* round1

gen fake1_1 = 2 - news1f

gen fake2_1 = 2 - news2f

gen fake3_1 = 2 - news3f

gen fake4_1 = 2 - news4f

gen fake5_1 = 2 - news5f

gen fake6_1 = 2 - news6f

gen fake7_1 = 2 - news7f

gen fake1 = (fake1_1 + fake2_1 + fake3_1 + fake4_1 + fake5_1 + fake6_1 + fake7_1)/7

gen fake1rep = (fake1_1 + fake3_1 + fake6_1)/3


* round 2

gen fake1_2 = 2 - w2_news1f

gen fake2_2 = 2 - w2_news2fb

gen fake3_2 = 2 - w2_news3f

gen fake4_2 = 2 - w2_news4fb

gen fake5_2 = 2 - w2_news5fb

gen fake6_2 = 2 - w2_news6f

gen fake7_2 = 2 - w2_news7fb

gen fake2 = (fake1_2 + fake2_2 + fake3_2 + fake4_2 + fake5_2 + fake6_2 + fake7_2)/7

gen fake2rep = (fake1_2 + fake3_2 + fake6_2)/3

* round 2 minus round 1

gen difffake = fake2 - fake1

gen difffakerep = fake2rep - fake1rep

* BELIEF IN PRO AND ANTI-BOLSONARO RUMORS IN SECOND ROUND

gen fake2antibolso = (fake5_2 + fake6_2)/2

gen fake2probolso = (fake1_2 + fake2_2 + fake4_2 + fake7_2)/4

* BELIEF IN REPEATED PRO AND ANTI-BOLSONARO RUMORS

gen difffakeantibolsorep = fake5_2 - fake5_1

gen difffakeprobolsorep = fake1_2 - fake1_1


* BELIEF IN REAL NEWS (1 is true, 2 is false)

* round1

gen true1_1 = 2 - news8t

gen true2_1 = 2 - news9t

gen true3_1 = 2 - news10t

gen true1 = (true1_1 + true2_1 + true3_1)/3

gen true1rep = true1_1

* round2

gen true1_2 = 2 - w2_news8t

gen true2_2 = 2 - w2_news9tb

gen true3_2 = 2 - w2_news10tb

gen true2 = (true1_2 + true2_2 + true3_2)/3

gen true2rep = true1_2

* round 2 minus round 1

gen difftrue = true2rep - true1rep


* DIFFERENCE IN TRUE AND FALSE

* round 1

gen truefalse1 = true1 - fake1

gen truefalse1rep = true1rep - fake1rep

* round 2

gen truefalse2 = true2 - fake2

gen truefalse2rep = true2rep - fake2rep

* round 2 minus round 1

gen difftruefalse = truefalse2 - truefalse1

gen difftruefalserep = truefalse2rep - truefalse1rep



*** OTHER OUTCOMES

* round 1

gen attention1_1 = (3 - attent1)/2

gen attention2_1 = (3 - attent2)/2

gen attention_1 = (attention1_1 + attention2_1)/2

gen ability1_1 = (4 - abil1)/2

gen ability2_1 = (4 - abil2)/2

gen ability_1 = (ability1_1 + ability2_1)/2

* round 2

gen attention1_2 = (3 - w2_attent1)/2

gen attention2_2 = (3 - w2_attent2)/2

gen attention_2 = (attention1_2 + attention2_2)/2

gen ability1_2 = (4 - w2_abil1)/2

gen ability2_2 = (4 - w2_abil2)/2

gen ability_2 = (ability1_2 + ability2_2)/2

* round 2 minus round 1

gen diff_attention = attention_2 - attention_1

gen diff_ability = ability_2 - ability_1

* mediatrust

gen mediatrust1 = mediatrust/10 

gen mediatrust2 = w2_mediatrust/10 

gen diff_mediatrust = mediatrust2 - mediatrust1


*** CONTROL VARIABLES (wave 1)

* sex

gen woman = gender

recode gender (1=1) (2=0) (else=.)

* age (0 to 1)

gen age01 = (age - 18)/62

* education

encode educ_cat, gen(education)

replace education = education - 1

* race (white)

encode race, gen(race_cat)

gen white = race_cat

recode white (2=1) (else=0)

* income

replace income  = (income - 1)/7

* religion (catholic and evangelical)

gen catholic = relig

recode catholic (1=1) (else=0)

gen evangelical = relig

recode evangelical (2=1) (else=0)

* political interest

replace interest = (4 - interest)/3

* past vote

gen pastbolso1 = pastvote

recode pastbolso1 (1=1) (else=0)

gen pasthaddad1 = pastvote

recode pasthaddad1 (2=1) (else=0)

gen pastbolso2 = w2_pastvote

recode pastbolso2 (1=1) (else=0)

gen pasthaddad2 = w2_pastvote

recode pasthaddad2 (2=1) (else=0)

* vote today

gen votebolso1 = votetoday

recode votebolso1 (1=1) (else=0)

gen votelula1 = votetoday

recode votelula1 (2=1) (else=0)

gen votebolso2 = w2_votetoday

recode votebolso2 (1=1) (else=0)

gen votelula2 = w2_votetoday

recode votelula2 (2=1) (else=0)

* left-right ideology

gen ideology1 = ideo/10

gen ideology2 = w2_ideo/10

* feeling thermometers

gen bolsofeeling1 = feel_1/10

gen lulafeeling1 = feel_2/10

gen morofeeling1 = feel_3/10

gen bolsofeeling2 = w2_feel_1/10

gen lulafeeling2 = w2_feel_2/10

gen morofeeling2 = w2_feel_3/10


**************************** ANALYSES

***** MAIN ITT AND CACE (without controls)
* (Codes for Table 2)


* ITT - all rumors in second round

reg fake2 treat, vce(robust)
eststo

reg true2 treat, vce(robust)
eststo

reg truefalse2 treat, vce(robust)
eststo


* CACE - all

ivreg2 fake2 (comply2=treat), first robust
eststo

ivreg2 true2 (comply2=treat), first robust
eststo

ivreg2 truefalse2 (comply2=treat), first robust
eststo


* ITT - repeat 

reg difffakerep treat, vce(robust)
eststo

reg difftrue treat, vce(robust)
eststo

reg difftruefalserep treat, vce(robust)
eststo


* CACE - repeat

ivreg2 difffakerep (comply2=treat), first robust
eststo

ivreg2 difftrue (comply2=treat), first robust
eststo

ivreg2 difftruefalserep (comply2=treat), first robust
eststo


* Table 2 (manually edited in the paper's tex file based on results) 


esttab using Table2.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear


**** comparing sizes of main ITTs for additive and repeated (not statistically different)

reg fake2 treat

est store model1

reg difffakerep treat

est store model2

suest model1 model2, vce(robust)

test [model1_mean]treat = [model2_mean]treat, accum


**** EFFECT SIZES (footnote 27)

* additive scale of rumor acceptance

reg fake2 treat, vce(hc3)

esizereg treat

* gain in scores for repeated rumor

reg difffakerep treat, vce(hc3)

esizereg treat

**** INTEGRATING STUDY 1 AND STUDY 2 (Footnote 28)

* effect size of ATE for study 1
scalar ATE1 =  -0.475239

*SE of effect size of Ate for study 1
scalar SEate1 =   0.075397 
scalar SEate1_2 = SEate1*SEate1
scalar invSEate1_2 = 1/SEate1_2


* effect size of ATE for study 2
scalar ATE2 = -0.145356

*SE of of effect size of Ate for study 2
scalar SEate2 = 0.077147
scalar SEate2_2 = SEate2*SEate2
scalar invSEate2_2 = 1/SEate2_2


* effect size of ATE for study 1 weighted
scalar ATE1w = ATE1*((invSEate1_2)/(invSEate1_2 + invSEate2_2))

* effect size of ATE for study 2 weighted
scalar ATE2w = ATE2*((invSEate2_2)/(invSEate1_2 + invSEate2_2))

* effect size of ATE pooled = 
scalar ATEpooled = ATE1w + ATE2w
display ATEpooled

* SE of effect size of ATE pooled
scalar SEATEpooled = sqrt((1)/(invSEate1_2 + invSEate2_2))
display SEATEpooled



* effect size of ATE pooled is -.31408145
* SE of effect size of ATE pooled is .0539218


**** DESCRIPTIVE STATISTICS
*(Codes for tables A16 and A17 in Appendix 14)

summarize fake1_1 fake2_1 fake3_1 fake4_1 fake5_1 fake6_1 fake7_1 fake1 true1_1 true2_1 true3_1 true1 treat comply2

summarize  fake1_2 fake2_2 fake3_2 fake4_2 fake5_2 fake6_2 fake7_2 fake2 true1_2 true2_2 true3_2 true2 treat comply2 if w2_consent!=.


**** BALANCE CHECKS
* (Codes for table A18 in Appendix 15)


* mean comparisons of pre-treatment covariates for treatment in wave 1

foreach var of varlist income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1 {

reg `var' treat, robust b

}

* joint effects in wave 1 (F-test)

reg treat income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1, robust 

test income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1

* joint effects in wave 1 (Chi-Square)

logit treat income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1, robust 

test income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1

* joint effects in wave 1 (Hotelling's T)

hotelling income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1, by(treat)


* mean comparisons of pre-treatment covariates for treatment in wave 2

foreach var of varlist income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1 {

reg `var' treat if select==1, robust b

}

* joint effects in wave 2 (F-test)

reg treat income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1 if select==1, robust 

test income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1

* joint effects in wave 2 (Chi-Square)

logit treatment income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1 if select==1, robust 

test income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1

* joint effects in wave 2 (Hotelling's T)

hotelling income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1 if select==1, by(treat)



***** PANEL ATTRITION
* (Codes for Table A19 in Appendix 16)


* mean comparisons

foreach var of varlist treat income education age01 woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 fake1rep true1 true1rep attention_1 ability_1 {

reg `var' dropped, robust b

}

reg dropped treat, robust

* F-test interactions

reg dropped i.treat##c.income i.treat##c.education i.treat##c.age01 i.treat##c.woman i.treat##c.catholic i.treat##c.evangelical i.treat##c.white i.treat##c.interest i.treat##c.pastbolso1 i.treat##c.pasthaddad1 i.treat##c.votebolso1 i.treat##c.votelula1 i.treat##c.bolsofeeling1 i.treat##c.lulafeeling1 i.treat##c.morofeeling1 i.treat##c.ideology1 i.treat##c.mediatrust1 i.treat##c.fake1 i.treat##c.fake1rep i.treat##c.true1 i.treat##c.true1rep i.treat##c.attention_1 i.treat##c.ability_1

testparm i.treat#c.income i.treat#c.education i.treat#c.age01 i.treat#c.woman i.treat#c.catholic i.treat#c.evangelical i.treat#c.white i.treat#c.interest i.treat#c.pastbolso1 i.treat#c.pasthaddad1 i.treat#c.votebolso1 i.treat#c.votelula1 i.treat#c.bolsofeeling1 i.treat#c.lulafeeling1 i.treat#c.morofeeling1 i.treat#c.ideology1 i.treat#c.mediatrust1 i.treat#c.fake1 i.treat#c.fake1rep i.treat#c.true1 i.treat#c.true1rep i.treat#c.attention_1 i.treat#c.ability_1



** MAIN ITT AND CACE (with controls)

** All rumors
* (Codes for Table A20 in Appendix 17)

* ITT

reg fake2 treat income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 true1 attention_1 ability_1, vce(robust)
eststo

reg true2 treat income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 true1 attention_1 ability_1, vce(robust)
eststo

reg truefalse2 treat income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 true1 attention_1 ability_1, vce(robust)
eststo

* CACE

ivreg2 fake2 income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 true1 attention_1 ability_1 (comply2=treat), first robust
eststo

ivreg2 true2 income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 true1 attention_1 ability_1 (comply2=treat), first robust
eststo

ivreg2 truefalse2 income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1 true1 attention_1 ability_1 (comply2=treat), first robust
eststo

* Table A20 (manually edited in the paper's tex file based on results) 


esttab using TableA20.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear


** Repeated rumors
* (Codes for Table A21 in Appendix 17)

* ITT 

reg difffakerep treat income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1rep true1rep attention_1 ability_1, vce(robust)
eststo

reg difftrue treat income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1rep true1rep attention_1 ability_1, vce(robust)
eststo

reg difftruefalserep treat income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1rep true1rep attention_1 ability_1, vce(robust)
eststo

* CACE

ivreg2 difffakerep income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1rep true1rep attention_1 ability_1 (comply2=treat), first robust
eststo

ivreg2 difftrue income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1rep true1rep attention_1 ability_1 (comply2=treat), first robust
eststo

ivreg2 difftruefalserep income education age woman catholic evangelical white interest pastbolso1 pasthaddad1 votebolso1 votelula1 bolsofeeling1 lulafeeling1 morofeeling1 ideology1 mediatrust1 fake1rep true1rep attention_1 ability_1 (comply2=treat), first robust
eststo

* Table A21 (manually edited in the paper's tex file based on results) 


esttab using TableA21.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear


*** BOUNDING
* (Codes for Table A22 in Appendix 18)
* notice that we use the terms upper and lower in the tables in the opposite direction as the code. Because our main expected effect is negative, we refer to increases in magnitude (more negative) as upper bounds and decreases in magnitude (more positive) as lower bounds

** Extreme bounds (Manski 1989)
*(upper bound: impute belief==0 for every dropped person in treatment, belief==1 for control)
*(lower bound: impute belief==1 for every dropped person in treatment, belief==0 for control)

* additive scale of fake, lower bound

gen fake2_lb = fake2
replace fake2_lb= 0 if fake2==. & treat==1
replace fake2_lb= 1 if fake2==. & treat==0

reg fake2_lb treat, vce(hc3)

* additive scale of fake, upper bound

gen fake2_ub = fake2
replace fake2_ub= 1 if fake2==. & treat==1
replace fake2_ub= 0 if fake2==. & treat==0

reg fake2_ub treat, vce(hc3)

* diff for false repeated rumors, lower bound 

gen difffake_lb = difffake
replace difffake_lb = -1 if difffake==. & fake1==1 & treat==1
replace difffake_lb = 0 if difffake==. & fake1==0 & treat==1
replace difffake_lb = 1 if difffake==. & fake1==1 & treat==0
replace difffake_lb = 1 if difffake==. & fake1==0 & treat==0

reg difffake_lb treat, vce(hc3)

* diff for false repeated rumors, upper bound 

gen difffake_ub = difffake
replace difffake_ub = 1 if difffake==. & fake1==1 & treat==1
replace difffake_ub = 1 if difffake==. & fake1==0 & treat==1
replace difffake_ub = 0 if difffake==. & fake1==1 & treat==0
replace difffake_ub = -1 if difffake==. & fake1==0 & treat==0

reg difffake_ub treat, vce(hc3)


** Trimming bounds (Lee 2009)

* additive scale, no tightening 

leebounds fake2 treat, select(select) cie

* gain scores for repeated false rumors, no tightening

leebounds difffake treat, select(select) cie


*** Magarlit & Shayo's lower

* additive scale, lower bound

gen fake2_lb2 = fake2
replace fake2_lb2= fake1 if fake2_lb2==.

reg fake2_lb2 treat, vce(hc3)

* gains scores for for repeated false rumors, lower bound

gen difffake_lb2 = difffake
replace difffake_lb2= 0 if difffake==. 

reg difffake_lb2 treat, robust


**** ADDITIONAL ANALYSES

*** MEASURES OF ABILITY AND ATTENTION


** ATTENTION (ITT and CACE)
* (Codes for Table A23 in Appendix 19)

reg attention_2 treat, vce(hc3)
eststo

ivreg2 attention_2 (comply2=treat), first robust
eststo

reg diff_attention treat, vce(hc3)
eststo

ivreg2 diff_attention (comply2=treat), first robust
eststo

* Table A23 (manually edited in the paper's tex file based on results) 

esttab using TableA23.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear



** ABILITY (ITT and CACE)
* (Codes for Table A24 in Appendix 19)

reg ability_2 treat, vce(hc3)
eststo

ivreg2 ability_2 (comply2=treat), first robust
eststo

reg diff_ability treat, vce(hc3)
eststo

ivreg2 diff_ability (comply2=treat), first robust
eststo

* Table A24 (manually edited in the paper's tex file based on results) 

esttab using TableA24.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear


** HTEs based on ability and attention
* (Codes for Table A25 in Appendix 19)

reg fake2 i.treat##c.ability_1, vce(hc3)
eststo

reg fake2 i.treat##c.attention_1, vce(hc3)
eststo

reg difffakerep i.treat##c.ability_1, vce(hc3)
eststo

reg difffakerep i.treat##c.attention_1, vce(hc3)
eststo

* Table A25 (manually edited in the paper's tex file based on results) 

esttab using TableA25.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear

****** HTEs (Education, Bolsonaro feeling, media trust, political interest)

** Additive scale of rumor acceptance
* (Codes for Table A26 in Appendix 20)

reg fake2 i.treat##c.education, vce(hc3)
eststo

reg fake2 i.treat##c.bolsofeeling1, vce(hc3)
eststo

reg fake2 i.treat##c.mediatrust1, vce(hc3)
eststo

reg fake2 i.treat##c.interest, vce(hc3)
eststo

* Table A26 (manually edited in the paper's tex file based on results) 

esttab using TableA26.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear

** Repeated rumors
* (Codes for Table A27 in Appendix 20)

reg difffakerep i.treat##c.education, vce(hc3)
eststo

reg difffakerep i.treat##c.bolsofeeling1, vce(hc3)
eststo

reg difffakerep i.treat##c.mediatrust1, vce(hc3)
eststo

reg difffakerep i.treat##c.interest, vce(hc3)
eststo

* Table A27 (manually edited in the paper's tex file based on results) 

esttab using TableA27.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear



*** Other DV: BELIEF IN FALSE PRO BOLSONARO (news1, news2, news4, and 7)
* (Codes for Table A28 in Appendix 21)

reg fake2probolso treat, vce(hc3)
eststo

ivreg2 fake2probolso (comply2=treat), first robust
eststo

reg difffakeprobolsorep treat, vce(hc3)
eststo

ivreg2 difffakeprobolsorep (comply2=treat), first robust
eststo

* Table A28 (manually edited in the paper's tex file based on results) 

esttab using TableA28.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear


*** Other DV: BELIEF IN FALSE ANTI BOLSONARO (news5, news6)
* (Codes for Table A29 in Appendix 21)

reg fake2antibolso treat, vce(hc3)
eststo

ivreg2 fake2antibolso (comply2=treat), first robust
eststo

reg difffakeantibolsorep treat, vce(hc3)
eststo

ivreg2 difffakeantibolsorep (comply2=treat), first robust
eststo

* Table A29 (manually edited in the paper's tex file based on results) 

esttab using TableA29.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear


*** Other DV: MEDIA TRUST
* (Codes for Table A30 in Appendix 21)

reg mediatrust2 treat, vce(hc3)
eststo

reg diff_mediatrust treat, vce(hc3)
eststo

ivreg2 mediatrust2 (comply2=treat), first robust
eststo

ivreg2 diff_mediatrust (comply2=treat), first robust
eststo


* Table A30 (manually edited in the paper's tex file based on results) 

esttab using TableA30.tex, star(* 0.05 ** 0.01 *** 0.001) se(2)
eststo clear



**** End of code
